NHATS Technical Paper #13

NATIONAL HEALTH AND AGING TRENDS STUDY (NHATS)

Calculating Work Productivity Loss in the National Study of Caregiving

Originally Published February 17, 2016
Revised August 27, 2019*

Suggested Citation: Wolff, Jennifer. 2016. Calculating Work Productivity Loss in the National Study of Caregiving. NHATS Technical Paper #13. Baltimore: Johns Hopkins University School of Public Health. Available at www.NHATS.org. The technical paper was prepared with funding from the National Institute on Aging (U01AG032947) and by the Office of the Assistant Secretary for Planning and Evaluation in the Department of Health and Human Services.
*Revisions to code are underlined

Introduction

Family and other unpaid caregivers who assist older adults for health and functioning reasons provide help with a range of activities such as self-care (e.g., eating, dressing, bathing, toileting), indoor or outdoor mobility, household activities (e.g., shopping for groceries, bills and banking, laundry), or managing health and health care activities (e.g, managing medications, skin care, dental care, talking with health professionals). Providing care may affect caregivers' own health and well-being in ways that are both positive and negative. For working caregivers, providing caregiving may affect their ability to be productive on the job, either due to missing time at work (absenteeism), or due to being distracted while being physically present at work (presenteeism). Prior studies find that more intense caregiving (e.g., helping with a greater number of tasks, providing more hours of care, or assisting an older adult with dementia) is associated with greater work productivity loss. (Giovannetti, 2009)

Measuring Work Productivity Loss

The Work Productivity and Activity Impairment (WPAI) questionnaire was originally developed by Margaret Reilly to quantitatively assess health-related work productivity loss for employed individuals. (Reilly, 1993). The measure explicitly considers time missed from work as well as reduced effectiveness on the job due to health problems by combining the amount of time missed from work with the amount of reduced productivity while at work. The measure has been validated for individuals with specific health problems such as panic disorder, asthma, irritable bowel disease, and gastroesophageal reflux disease. (See Giovennetti, 2009 and www.reillyassociates.net)

In a prior study, the WPAI questionnaire was modified to reflect caregiving-related work productivity and activity impairment. (Giovannetti, 2009). Caregivers who report that they are working for pay are asked questions about the impact of caregiving on productivity in the previous week. In the originally published instrument, questions delineated lost time and productivity due to caregiving responsibilities using a 7 day recall period as follows:

- absenteeism is measured as a percentage equal to [hours missed from work due to caregiving/(hours. missed due to caregiving + hours actually worked)] * 100;
- presenteeism is defined as the degree to which caregiving affected productivity while at work [(measured as a number on a scale from 1–10)/10] * 100;
- productivity loss is measured as a percentage equal to [absenteeism + (% of time worked * presenteeism)] * 100;

Work Productivity Loss in the National Study of Caregiving (NSOC)

The approach to determining work productivity loss in the NSOC (described in Wolff et al. 2016) was modified to incorporate slight differences in question wording, recall period, and response categories. Caregivers who indicated that they were working for pay were asked the number of hours that they "typically" work, as well as the number of hours "actually" worked in the past week. The hours "typically worked" was used as the denominator in the calculation of absenteeism, because NSOC asks caregivers to report time missed from work due to caregiving using a one-month recall period rather

than a one-week recall period. Numbers of hours worked (which is asked using a one-week reference period) were therefore also standardized to a one-month period by multiplying by 4.33. For caregivers who reported that they worked variable schedules (n=53 in 2011), we used hours worked in the last week, or if not reported, the reported hours worked during last week they had worked.

For those caregivers who reported that they missed days of work due to caregiving (n=16 in 2011), we assumed "something less than 8 hours" was equivalent to 6 hours of time missed from work, "something more than 8 hours" was equivalent to 10 hours of time missed from work, and that "about 8 hours" was 8 hours of time per day of missed work. For caregivers who reported that they missed work due to caregiving but for whom information regarding the unit of time missed from work was not provided (n=4 in 2011), we assumed that they missed 14 hours from work in the past month as this was the average time missed from work due to caregiving among working caregivers who reported missing any work due to caregiving.

SAS code to calculate work productivity loss in the NSOC is presented in Appendix A. The variables that are used to calculate work productivity loss in the NSOC are presented in Appendix B.

References

Giovannetti E, Wolff J, Frick K, Boult C. Validation of the Work Productivity and Activity Impairment Questionnaire for Informal Caregivers (WPAI:CG) *Value in Health*. 2009;12(6):1011-7

Reilly M, Zbrozek A, Dukes E. The validity and reproducibility of a work productivity and activity impairment instrument. Pharmacoeconomics 1993;4(5):353-65.

Wolff JL, Spillman BC, Freedman VA, Kasper JD. A National Profile of Family and Unpaid Caregivers Who Assist Older Adults With Health Care Activities. *JAMA Intern Med*. Published online February 15, 2016. doi:10.1001/jamainternmed.2015.7664

Appendix A. SAS code to calculate work productivity loss in the National Study of Caregiving (NSOC)

```
CGWORK4PAY=.;
IF cec1wrk4pay=1 THEN CGWORK4PAY=1;
ELSE CGWORK4PAY=0;

MISSEDWK=0;
IF cec1msswrkmt=1 then MISSEDWK=1;

CGMISSEDWK=0;
IF cec1msswrkmt=1 and cec1mswkhelp=1 then CGMISSEDWK=1;
else cgmissedwk=0;

MISSEDHRS=0;
IF cec1mswkunit=1 and cec1mswkhlph>0 then MISSEDHRS=cec1mswkhlph;
else if cec1mswkunit=2 and cec1mswkhlpd>0 and cec1wkhrperd=1 then
MISSEDHRS=(cec1mswkhlpd*8);
else if cec1mswkunit=2 and cec1mswkhlpd>0 and cec1wkhrperd=3 then
MISSEDHRS=(cec1mswkhlpd*10);
else if cec1mswkunit=2 and cec1mswkhlpd>0 and cec1wkhrperd=2 then
MISSEDHRS=(cec1mswkhlpd*6);
else if cec1mswkunit=2 and cec1mswkhlpd>0 and cec1wkhrperd<0 then
MISSEDHRS=(cec1mswkhlpd*8); /*DON'T KNOW HOW MANY HOURS/DAY*/
else if CGMISSEDWK=1 and cec1mswkunit<0 then MISSEDHRS=14; /*CODE TO AVERAGE
MISSED HOURS AMONG WORKING CAREGIVERS WHO MISSED TIME DUE TO CAREGIVING*/

HOURSWORKED=.;
IF 90>CEC1HRSWEEK>0 THEN HOURSWORKED=CEC1HRSWEEK*4.33;
ELSE if cec1hrslstwk>0 then HOURSWORKED=cec1hrslstwk*4.33;
ELSE if cec1hrslast>0 then HOURSWORKED= cec1hrslast *4.33;
ELSE HOURSWORKED=0;

MISSEDHRS_RC=.;
IF MISSEDHRS>HOURSWORKED THEN MISSEDHRS_RC=HOURSWORKED;
ELSE MISSEDHRS_RC=MISSEDHRS;

ABSENTEEISM=0;
IF CGMISSEDWK=1 AND MISSEDHRS_RC>0 THEN
ABSENTEEISM=(MISSEDHRS_RC/(HOURSWORKED));
else absenteeism=0;

PRESENTEEISM=0;
IF cec1hlpafwk1=1 then PRESENTEEISM=(cec1hlpafwk2/10);
else PRESENTEEISM=0;

WPAI=ABSENTEEISM+((1-ABSENTEEISM)*PRESENTEEISM);
IF WPAI>1 THEN WPAI=1;
```

Appendix B. Sources of information for Work Productivity Loss in the National Study of Caregiving (NSOC)

Item	Variable Name	Variable Label
EC1	cec1wrk4pay	C1 EC1 WORK FOR PAY LAST WEEK
EC7	cec1hrsweek	C1 EC7 HRS PER WEEK AT MAIN JOB
EC8	cec1hrslstwk	C1 EC8 NUMB HRS WORKED LAST WEEK
EC9	cec1hrslast	C1 EC9 HRS WRKED LST WEEK WORKED
EC13	cec1msswrkmt	C1 EC13 EVER ABSENT FROM WORK
EC14	cec1mswkhelp	C1 EC14C MISS WORK HELPING
EC15	cec1mswkunit	C1 EC15 HRS MISS WORK HELP UNIT
EC15a	cec1mswkhlph	C1 EC15A NUM HRS MISS WORK HELP
EC15b	cec1mswkhlpd	C1 EC15B NUM DYS MISS WORK HELP
EC16	cec1wkhrperd	C1 EC16 NUM OF HOURS IN WORK DAY
EC17	cec1hlpafwk1	C1 EC17 HELPING AFFECTS WORK